```
> restart:
```

Leçon 4 : Intégration basique.

```
> int(1/(1+x^2),x=0..1);
```

$$\frac{1}{4}\pi$$

```
> int(1/(1+x^2),x);
```

$$\arctan(x)$$

De toutes façons vous finirez par appeler une intégrale I, qui est l'imaginaire pur bien connu, et vous aurez droit à ceci !

```
> I := int(1/(1+x^2),x);
Error, Illegal use of an object as a name
```

Maple accepte les intervalles non bornés.

```
> int(exp(-x^2/2),x=-infinity..infinity);
```

$$\sqrt{2}\,\sqrt{\pi}$$

Maple possède une bibliothèque de fonctions qui s'expriment à l'aide d'intégrales...et ne manque pas de s'en servir (faire help Ei pour plus de détails...)

```
> int(exp(x)/x,x);
```

$$-\mathrm{Ei}(1, -x)$$

```
> int(exp(x)/x,x=1..2); evalf(");
```

$$-\mathrm{Ei}(1, \text{-}2) + \mathrm{Ei}(1, \text{-}1)$$

$$3.059116540$$

```
> f := 1/(x^4+x^2+1) : int(f,x);
```

$$\frac{1}{4}\ln(x^2+x+1)+\frac{1}{6}\sqrt{3}\arctan\left(\frac{1}{3}(2x+1)\sqrt{3}\right)-\frac{1}{4}\ln(x^2-x+1)+\frac{1}{6}\sqrt{3}\arctan\left(\frac{1}{3}(2x-1)\sqrt{3}\right)$$

A titre de vérification, voici comment obtenir la décomposition en éléments simples (ce qui n'est pas évident à dénicher dans la documentation)

```
> convert(f,parfrac,x);
```

$$-\frac{1}{2}\frac{-1+x}{x^2-x+1}+\frac{1}{2}\frac{1+x}{x^2+x+1}$$

```
> int(",x);
```

$$\frac{1}{4}\ln(x^2+x+1)+\frac{1}{6}\sqrt{3}\arctan\left(\frac{1}{3}(2x+1)\sqrt{3}\right)-\frac{1}{4}\ln(x^2-x+1)+\frac{1}{6}\sqrt{3}\arctan\left(\frac{1}{3}(2x-1)\sqrt{3}\right)$$

Décomposer sur C est plus acrobatique, si on ne connait pas les facteurs : le quatrième paramètre indique que les facteurs doivent être de la forme a+ib*sqrt(3). Il ne semble pas que Maple factorise de lui-même sur C, sauf numeriquement.

```
> convert(f,parfrac,x,(-3)^(1/2));convert(f,parfrac,x,complex);
```

$$-\frac{1}{6}\frac{-3+I\sqrt{3}}{2x+1-I\sqrt{3}}+\frac{1}{6}\frac{3+I\sqrt{3}}{2x+1+I\sqrt{3}}+\frac{1}{6}\frac{-3+I\sqrt{3}}{2x-1+I\sqrt{3}}-\frac{1}{6}\frac{3+I\sqrt{3}}{2x-1-I\sqrt{3}}$$

$$\frac{.2499999999+.1443375672\,I}{x+.5000000000+.8660254038\,I}+\frac{.2499999999-.1443375672\,I}{x+.5000000000-.8660254038\,I}$$
$$+\frac{-.2499999999+.1443375672\,I}{x-.5000000000+.8660254038\,I}+\frac{-.2499999999-.1443375672\,I}{x-.5000000000-.8660254038\,I}$$

Il n'y a pas réellement d'intégrales doubles dans Maple, mais on peut "Fubiniser".

```
> f := x^2+x*y:int(int(f,y=0..x),x=0..1);
```

$$\frac{3}{8}$$

Maple sait calculer symboliquement avec les intégrales

```
> an := int(phi(x)*cos(n*x),x=-Pi..Pi)/Pi;
```

$$an := \frac{\int_{-\pi}^{\pi} \phi(x)\cos(n\,x)\,dx}{\pi}$$

```
> phi := x->abs(x):an;
```

$$2\,\frac{\cos(\pi\,n) - 1 + n\,\sin(\pi\,n)\,\pi}{n^2\,\pi}$$

```
> assume(n, integer); an;
```

$$2\,\frac{(-1)^{n\sim} - 1}{n\sim^2\,\pi}$$

Forme inerte, integration par parties, changement de variable.

Maple permet de définir une forme *inerte* (ou *symbolique*) pour l'intégrale, qui correspond vaguement à l'idée de primitive, i.e. une expression non évaluée directement mais ayant certaines propriétés...

```
> I1 := Int(1/(1+x^2),x);diff(I1,x);series(I1,x,8);
```

$$I1 := \int \frac{1}{1+x^2}\,dx$$

$$\frac{1}{1+x^2}$$

$$x - \frac{1}{3}x^3 + \frac{1}{5}x^5 - \frac{1}{7}x^7 + \mathrm{O}(x^9)$$

I1 est une integrale "symbolique", que l'on peut néanmoins évaluer, si cela est possible, non pas avec l'instruction eval mais en utilisant l'instruction value .

```
> eval(I1);
```

$$\int \frac{1}{1+x^2}\,dx$$

```
> value(I1);
```

$$\arctan(x)$$

Les calculs "classiques" sont définis, entre autres, dans le package student, mais ne s'appliquent généralement qu'à la forme inerte.

```
> with(student);
```

[D, *Diff*, *Doubleint*, *Int*, *Limit*, *Lineint*, *Product*, *Sum*, *Tripleint*, *changevar*, *completesquare*, *distance*, *equate*, *integrand*, *intercept*, *intparts*, *leftbox*, *leftsum*, *makeproc*, *middlebox*, *middlesum*, *midpoint*, *powsubs*, *rightbox*, *rightsum*, *showtangent*, *simpson*, *slope*, *summand*, *trapezoid*]

```
> I2 := Int(t/(1+t^2)/sqrt(1-t^4),t=0..1);
```

$$I2 := \int_0^1 \frac{t}{(1+t^2)\sqrt{1-t^4}}\, dt$$

```
> value(I2);
```

$$\frac{1}{2} + \left(\sum_{\_\alpha = \mathrm{RootOf}(\_Z^2+1)} \left(\frac{1}{8}\sqrt{2}\ \_\alpha\ \mathrm{EllipticPi}\left(\frac{1}{2}, \frac{1}{2}\sqrt{2} \right) \right) \right)$$

Intéressant, mais peu pratique : cela doit pouvoir s'arranger...

```
> evalf(");
```

$$.5000000000$$

On va donc mâcher le travail de Maple.

```
> I3 := changevar(u=t^2,I2,u);
```

$$I3 := \int_0^1 \frac{1}{2} \frac{1}{(1+u)\sqrt{1-u^2}}\, du$$

```
> I4 := changevar(u=cos(x),I3,x);
```

$$I4 := \int_0^{1/2\,\pi} \frac{1}{2} \frac{\sin(x)}{(1+\cos(x))\sqrt{1-\cos(x)^2}}\, dx$$

C'est là que cela se gâte...

```
> I5 := simplify(I4,symbolic);
```

$$I5 := \frac{1}{2} \int_0^{1/2\,\pi} \frac{1}{1+\cos(x)}\, dx$$

```
> value(");
```

$$\frac{1}{2}$$

On peut remarquer que le fait d'avoir une intégrale impropre ne gêne aucunement Maple.

```
> f := 1/(1+x^2)^n:In := Int(f,x);
```

$$In := \int \frac{1}{(1+x^2)^{n\sim}}\, dx$$

```
> Jn:= intparts(In,f);
```

$$Jn := \frac{x}{(1+x^2)^{n\sim}} - \int -2 \frac{n\sim x^2}{(1+x^2)^{n\sim}\,(1+x^2)}\, dx$$

```
> simplify(Jn);
```

$$(1+x^2)^{(-n\sim)}\, x + 2\, n\sim \int (1+x^2)^{(-n\sim-1)}\, x^2\, dx$$

```
> restart:
```

Séries etc...

Diverses manières de travailler avec les séries, applications aux séries de Fourier.
D'abord les séries de "Taylor".

```
> f1 := sqrt(1+x+x^2);
```

$$f1 := \sqrt{1+x+x^2}$$

```
> series(f1,x=0,4);
```
le dernier paramètre est l'ordre désiré.

$$1+\frac{1}{2}x+\frac{3}{8}x^2-\frac{3}{16}x^3+\frac{3}{128}x^4+\frac{15}{256}x^5+O(x^6)$$

```
> series(f1,x=0,8);
```

$$1+\frac{1}{2}x+\frac{3}{8}x^2-\frac{3}{16}x^3+\frac{3}{128}x^4+\frac{15}{256}x^5-\frac{57}{1024}x^6+\frac{21}{2048}x^7+\frac{867}{32768}x^8-\frac{1893}{65536}x^9+O(x^{10})$$

```
> Order := 6:
```
on peut aussi utiliser la variable Order

```
> ee := series(exp(x),x);
```
x sans indication sgnifie x=0

$$ee := 1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5+O(x^6)$$

```
> series(n!,n=infinity,3);
```

$$\frac{\dfrac{\sqrt{2}\sqrt{\pi}}{\sqrt{\dfrac{1}{n}}}+\dfrac{1}{12}\sqrt{2}\sqrt{\pi}\sqrt{\dfrac{1}{n}}+\dfrac{1}{288}\sqrt{2}\sqrt{\pi}\left(\dfrac{1}{n}\right)^{(3/2)}+O\left(\left(\dfrac{1}{n}\right)^{(5/2)}\right)}{\left(\dfrac{1}{n}\right)^{n}\mathbf{e}^{n}}$$

Maple connait le formule de Stirling...et sait faire des developpements génralisés !

```
> p := convert(ee,polynom);
```
pour récuperer la partie régulière

$$p := 1+x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5$$

```
> solve(ee=y,x);
```
#on peut même calculer le developpement d'une fonction réciproque ...
y doit être une variable libre !

$$y-1-\frac{1}{2}(y-1)^2+\frac{1}{3}(y-1)^3-\frac{1}{4}(y-1)^4+\frac{1}{5}(y-1)^5+O((y-1)^6)$$

add est utilisé pour faire des sommations (plus efficaces que les boucles ...)

```
> add(1/k,k=1..30);
```

$$\frac{9304682830147}{2329089562800}$$

```
> add(x^k/k!,k=1..10);
```

$$x+\frac{1}{2}x^2+\frac{1}{6}x^3+\frac{1}{24}x^4+\frac{1}{120}x^5+\frac{1}{720}x^6+\frac{1}{5040}x^7+\frac{1}{40320}x^8+\frac{1}{362880}x^9+\frac{1}{3628800}x^{10}$$

```
> add(k^2,k=1..n);
Error, unable to execute add
```

sum est utilisé pour faire des sommes formelles ...

```
> sum(k^2,k=1..n);
```

$$\frac{1}{3}(n+1)^3 - \frac{1}{2}(n+1)^2 + \frac{1}{6}n + \frac{1}{6}$$

```
> sum(x^k/k!,k=1..infinity);
```
et même avec une infinité de termes.

$$\mathbf{e}^x (1 - \mathbf{e}^{(-x)})$$

On peut aussi étudier la convergence des séries :

```
> sum((-1)^n/((-1)^n+sqrt(n)), n=2..infinity);# Maple sèche...mais
  pas nous!
```

$$\sum_{n=2}^{\infty} \frac{(-1)^n}{(-1)^n + \sqrt{n}}$$

```
> DA:=asympt(epsilon/(epsilon+sqrt(n)),n,2);# donc elle diverge
```

$$DA := \varepsilon \sqrt{\frac{1}{n}} - \frac{\varepsilon^2}{n} + \varepsilon^3 \left(\frac{1}{n}\right)^{(3/2)} + O\left(\frac{1}{n^2}\right)$$

```
> sum((-1)^n/ln((-1)^n+n),n=2..infinity):DA:=asympt(epsilon/ln(eps
  ilon+n),n,2);# donc elle converge
```

$$DA := \frac{\varepsilon}{\ln(n)} - \frac{\varepsilon^2}{\ln(n)^2\, n} + O\left(\frac{1}{n^2}\right)$$

On peut aussi utiliser Maple et pour regarder quel type de singularité on a (fonctions holomorphes)

```
> series(1/sin(x), x, 5);type(", laurent);
```

$$x^{-1} + \frac{1}{6}x + O(x^3)$$

true

```
> series(ln(x+x^2), x, 3);type(", laurent);
```

$$\ln(x) + x - \frac{1}{2}x^2 + O(x^3)$$

false

Et aussi récupérer des coefficients du dévellopement en série et donc le résidu ....:

```
> b := series(tan(x)/(x*sin(x)), x=0, 13);
```

$$b := x^{-1} + \frac{1}{2}x + \frac{5}{24}x^3 + \frac{61}{720}x^5 + \frac{277}{8064}x^7 + \frac{50521}{3628800}x^9 + O(x^{11})$$

```
> op(b);#op(2i-1,b) coefficient de x à la puissance op(2i,b)
```

$$1, -1, \frac{1}{2}, 1, \frac{5}{24}, 3, \frac{61}{720}, 5, \frac{277}{8064}, 7, \frac{50521}{3628800}, 9, O(1), 11$$

Applications aux séries de Fourier. On va développer |x| .

```
> restart:
> a := n -> int(f(t)*cos(n*t)/Pi,t=-Pi..Pi);
```

$$a := n \to \int_{-\pi}^{\pi} \frac{f(t)\cos(n\,t)}{\pi}\, dt$$

```
> b := n -> int(f(t)*sin(n*t)/Pi,t=-Pi..Pi);
```

$$b := n \to \int_{-\pi}^{\pi} \frac{f(t)\sin(n\,t)}{\pi}\,dt$$

```
> f := t -> abs(t);
```

$$f := \mathrm{abs}$$

```
> a(11);b(11);
```

$$-\frac{4}{121}\frac{1}{\pi}$$

$$0$$

```
> S := (n,x) -> a(0)/2+add(a(i)*cos(i*x)+b(i)*sin(i*x),i=1..n);
Warning, `i` in call to `add` is not local
```

$$S := (n, x) \to \frac{1}{2}\,\mathrm{a}(0) + \mathrm{add}(\mathrm{a}(i)\cos(i\,x) + \mathrm{b}(i)\sin(i\,x),\ i = 1\ ..\ n)$$

```
> S(3,x);
>
```

$$\frac{1}{2}\pi - 4\frac{\cos(x)}{\pi} - \frac{4}{9}\frac{\cos(3\,x)}{\pi}$$

```
> S(5,x);
```

$$\frac{1}{2}\pi - 4\frac{\cos(x)}{\pi} - \frac{4}{9}\frac{\cos(3\,x)}{\pi} - \frac{4}{25}\frac{\cos(5\,x)}{\pi}$$

```
> value(a(n));
```

$$2\,\frac{\cos(\pi\,n) + \pi\,n\sin(\pi\,n)}{n^2\,\pi} - \frac{2}{n^2\,\pi}$$

```
> assume(n, integer);
> value(a(n));
```

$$2\,\frac{(-1)^{n\sim}}{n\sim^2\,\pi} - \frac{2}{n\sim^2\,\pi}$$

```
> for k in [1,2,5,10,20] do plot(evalf(S(k,x)),x=-Pi..Pi); od;
```